// File: lib/openzeppelin-contracts/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: lib/openzeppelin-contracts/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: lib/openzeppelin-contracts/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: src/EtherCapsuleAuction.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.20;

import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";
import {ReentrancyGuard} from "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import {IERC721} from "@openzeppelin/contracts/token/ERC721/IERC721.sol";

error InputMismatch();
error AuctionNotStarted();
error MaxPerMintExceeded();
error InsufficientPrice();
error WithdrawFailed();
error InvalidCurrentTokenId();
error RefundAlreadyProcessed();
error AuctionNotOver();
error InvalidBid();
error RefundFailed(uint256 tokenId);
error OnlyEOA();

contract EtherCapsuleAuction is Ownable, ReentrancyGuard {
    IERC721 public immutable ETHER_CAPSULE;

    struct AuctionParams {
        uint64 startPrice;
        uint64 endPrice;
        uint64 decay;
        uint32 step;
        uint32 startTime;
    }

    struct Bid {
        address bidder;
        uint64 amount;
        bool refundProcessed;
    }

    AuctionParams public auctionParams;
    mapping(uint256 => Bid) public tokenBids;
    // Initial value is the first token minted to this contract
    uint256 public currentTokenId = 2075;
    uint256 public settlePrice;

    constructor(
        address _capsuleAddress,
        uint64 _startPrice,
        uint64 _endPrice,
        uint64 _decay,
        uint32 _step,
        uint32 _startTime
    ) {
        ETHER_CAPSULE = IERC721(_capsuleAddress);

        auctionParams = AuctionParams(_startPrice, _endPrice, _decay, _step, _startTime);
    }

    function airdrop(address[] calldata receivers, uint16[] calldata amounts) external onlyOwner {
        if (receivers.length != amounts.length) revert InputMismatch();

        unchecked {
            for (uint256 i; i < receivers.length; ++i) {
                address receiver = receivers[i];
                uint16 amount = amounts[i];
                uint256 _currentTokenId = currentTokenId;

                // Loop through and transfer tokens
                for (uint16 j; j < amount; ++j) {
                    ETHER_CAPSULE.transferFrom(address(this), receiver, _currentTokenId + j);
                }

                currentTokenId += amount;
            }
        }
    }

    modifier onlyEOA() {
        if (tx.origin != msg.sender) revert OnlyEOA();
        _;
    }

    // =========================================================================
    //                            Auction Functions
    // =========================================================================

    function getAuctionPrice() public view returns (uint256) {
        AuctionParams memory _auctionParams = auctionParams;
        if (_auctionParams.startTime == 0) return _auctionParams.startPrice;
        uint256 elapsed = block.timestamp - _auctionParams.startTime;
        uint256 steps = elapsed / _auctionParams.step;
        uint256 decay = steps * _auctionParams.decay;
        unchecked {
            if (decay > _auctionParams.startPrice - _auctionParams.endPrice) {
                return _auctionParams.endPrice;
            }
            return _auctionParams.startPrice - decay;
        }
    }

    function auctionMint(uint256 qty) external payable nonReentrant onlyEOA {
        if (auctionParams.startTime == 0) revert AuctionNotStarted();
        if (qty > 3) revert MaxPerMintExceeded();

        uint256 currentPrice = getAuctionPrice();
        uint256 totalPrice = currentPrice * qty;
        if (msg.value < totalPrice) revert InsufficientPrice();

        // Record bids for each token id
        uint256 _currentTokenId = currentTokenId;
        for (uint256 i; i < qty;) {
            uint256 tokenId = _currentTokenId + i;
            tokenBids[tokenId] = Bid(msg.sender, uint64(currentPrice), false);
            ETHER_CAPSULE.transferFrom(address(this), msg.sender, tokenId);
            unchecked {
                ++i;
            }
        }

        unchecked {
            // Increment current token id
            currentTokenId += qty;

            // Refund if over
            uint256 difference = msg.value - totalPrice;
            if (difference > 0) {
                _transferETH(msg.sender, difference);
            }
        }
    }

    function processRefunds(uint256[] calldata tokenIds) external onlyOwner nonReentrant {
        if (settlePrice == 0) revert AuctionNotOver();

        for (uint256 i; i < tokenIds.length;) {
            uint256 tokenId = tokenIds[i];
            Bid memory bid = tokenBids[tokenId];

            if (bid.bidder == address(0)) revert InvalidBid();
            if (bid.refundProcessed) revert RefundAlreadyProcessed();

            tokenBids[tokenId].refundProcessed = true;
            uint256 priceDifference = bid.amount - settlePrice;
            if (priceDifference > 0) {
                bool success = _transferETH(bid.bidder, priceDifference);
                if (!success) revert RefundFailed(tokenId);
            }

            unchecked {
                ++i;
            }
        }
    }

    // =========================================================================
    //                           Owner Only Functions
    // =========================================================================

    function setCurrentTokenId(uint256 _currentTokenId) external onlyOwner {
        if (ETHER_CAPSULE.ownerOf(_currentTokenId) != address(this)) {
            revert InvalidCurrentTokenId();
        }
        currentTokenId = _currentTokenId;
    }
    
    function setSettlePrice(uint256 _settlePrice) external onlyOwner {
        settlePrice = _settlePrice;
    }

    function setAuctionStart(uint32 _startTime) external onlyOwner {
        auctionParams.startTime = _startTime;
    }

    function setAuctionParams(uint64 _startPrice, uint64 _endPrice, uint64 _decay, uint32 _step, uint32 _startTime)
        external
        onlyOwner
    {
        // Passing a _startTime > 0 will set the auction start time
        auctionParams = AuctionParams(_startPrice, _endPrice, _decay, _step, _startTime);
    }

    function withdrawFunds(address receiver) external onlyOwner {
        (bool sent,) = receiver.call{value: address(this).balance}("");
        if (!sent) {
            revert WithdrawFailed();
        }
    }

    function _transferETH(address to, uint256 value) internal returns (bool) {
        (bool success,) = to.call{value: value, gas: 30000}(new bytes(0));
        return success;
    }
}